CHAPTER 5

Circuit Elements

A	Heterostructure field-effect transistors (HFETs)	SPICE3
B	Non-linear dependent sources	SPICE3
B	GaAs field-effect transistor	PSPICE
C	Capacitors	
D	Diodes	
E	Linear voltage-controlled voltage sources	
F	Linear current-controlled current sources	
G	Linear voltage-controlled current sources	
H	Linear current-controlled voltage sources	
I	Independent current sources	
J	Junction field-effect transistors (JFETs)	
K	Coupled inductors (transformers)	
L	Inductors	
M	Metal oxide semiconductor field-effect transistors (MOSFETs)	
N	Heterojunction bipolar transistors (HBTs)	SPICE3
O	Lossy transmission lines (LTRA)	SPICE3
Q	Bipolar junction transistors (BJTs)	
R	Resistors	
S	Voltage-controlled switches	
T	Lossless transmission lines	
U	Uniform distributed *RC* lines (URC)	SPICE3
U	Digital devices	PSPICE
V	Independent voltage sources	
W	Current-controlled switches	
X	Subcircuit call	
Z	GaAs metal semiconductor field-effect transistors (MESFETs)	SPICE3

A circuit's topology is described by listing all circuit elements and specifying the nodes to which they are connected. Each statement describes one circuit element. Statements always start with the element name. The node numbers are then listed, and the element value or model name follows. Optional parameters may also follow the listed nodes. The name of an element always starts with a specific letter which indicates the type of element. Table 5.1 shows the list of elements and their code letters. Subsequent letters are reserved for user-defined element names. In element description and in examples SPICE keywords and letters are written using **bold** characters. Names that can be chosen by the user start with capital letters and *italic* characters are used.

For example, a capacitor name must begin with the letter **C** and can have up to eight characters (letters or digits): **C**2, **C**coupl, **C**27, and **C**BLOCK. The original SPICE2 version, which was written in Fortran, requires the use of capital letters. Most current SPICE programs allow both small and capital letters. In most SPICE versions, nodes numbers must be nonnegative integers but need not be numbered sequentially. In a newer SPICE programs, node can also have names described by strings of letters and digits. The number zero is always reserved for ground and must be used that way. Each node in the circuit must have a dc path to ground. Every node must have at least two connections, except for transmission line nodes (to permit unterminated transmission lines) and MOSFET substrate nodes (which actually have two internal connections).

B - Nonlinear Dependent Source SPICE3 only



General form	*Restrictions*
B*name Pnode Nnode* [**I**=*Expression*] [**V**=*Expression*]	SPICE3

Examples	
BN 3 4 **I**=**COS**(V(8)+**SIN**(**V**(13))	SPICE3
B1 7 8 **V**=**ln**(**cos**(**log**(**V**(3,5)^2)))-**V**(8)^3+**V**(7)^**V**(8)	SPICE3
B7 12 21 **I**=1mA	SPICE3
BNEW 3 7 **V**=**exp**(2***I**(VS3))	SPICE3

The nonlinear source must begin with the letter **B**. *Pnode* and *Nnode* are the positive and negative nodes, respectively. The values of the **V** and **I** parameters determine the voltages and currents across and through the device, respectively. There is no distinction between current-controlled and voltage-controlled sources for the **B** element. If **I**= is given, then the device is a current source. If **V**= is given, the device is a voltage source. The small-signal ac behavior of the **B** source is a linear dependent source with a gain constant equal to the derivative (or derivatives) of the source at the dc operating point.

Expression may be any function of voltages and currents through voltage sources in the system. The source output, determined by the **V**= or **I**= option, can be either a voltage or a current. Function defined in SPICE3 program are shown in Table 5.1.

Table 5.1 Functions defined in SPICE3

Function	*Description*	*Units*
abs(x)	absolute value	
acos(x)	arccosine	result in radians
acosh(x)	inverse hyperbolic cosine	-
asin(x)	arcsine	result in radians
asinh(x)	inverse hyperbolic sine	-
atan(x)	arctan	result in radians
atanh(x)	inverse hyperbolic tangent	-
cos(x)	cosine	x in radians
cosh(x)	hyperbolic cosine	-
exp(x)	exponential function	-
ln(x)	logarithm with base e	-
log(x)	logarithm with base 10	-
sin(x)	sine	x in radians
sinh(x)	hyperbolic sine	-
sqrt(x)	square root	-
tan(x)	tangent	x in radians
u(x)	$u(x) = \begin{cases} 1 & x > 0 \\ 0 & x < 0 \end{cases}$	-
uramp(x)	$u(x) = \begin{cases} x & x > 0 \\ 0 & x < 0 \end{cases}$	

The following operations are defined: + - * / ^ unary -

If the argument of **log**, **ln**, or **sqrt** becomes less than zero, the absolute value of the argument is used. If a divisor becomes zero or the argument of **log** or **ln** becomes zero, an error will result. Other problems may occur when the argument for a function in a partial derivative enters a region where that function is undefined.

To get time into an expression, you can integrate the current from a constant current source with a capacitor and use the resulting voltage to represent time. Remember to set the initial voltage across the capacitor and use **UIC** in the **.TRAN** statement.

B - GaAs FET PSPICE only

GASFET



General forms	*Restrictions*
B*name Dnode Gnode Snode Model* [*Rarea*]	PSPICE
Examples	
B8 2 3 5 BMOD	PSPICE
B3 1 2 4 BMODEL 0.5	PSPICE

A GaAsFET is described by a statement that starts with the name of the GaAs FET device **B***name*. This name must start with the letter **B**. Node numbers *Dnode*, *Gnode*, and *Snode* for drain, gate, and source follows the name. Next, the model *Model* name is listed. Model parameters are specified in the **.MODEL** statement. *Rarea* is the relative area factor. If *Rarea* is not specified, 1 is assumed.

1. GaAs FET model

.MODEL *Model_name* **GASFET** [*Model parameters*]

2. Model parameters

In PSPICE, four different models are implemented: level 1 through level 4.

Parameters for All Levels				
Name	*Parameter*	*Units*	*Default*	*Typical*
LEVEL	Model index	-	1	2
VTO	Pinch-off voltage	V	-2.5	-2.0
BETA	Transconductance coefficient	A/V^2	0.1	0.1
LAMBDA	Channel-length modulation parameter	1/V	0	10^3

RD	Drain ohmic resistance	Ω	0	100
RS	Source ohmic resistance	Ω	0	100
RG	Gate ohmic resistance	Ω	0	10
IS	Gate p-n saturation current	A	10-^{14}	10^{-14}
N	Gate p-n emission coefficient	–	1	1.2
VBI	Gate p-n potential	V	1.0	0.9
CGS	Zero-bias G-S junction capacitance	F	0	5 pF
CGD	Zero-bias G-D junction capacitance	F	0	5 pF
CDS	Zero-bias D-S capacitance	F	0	1 pF
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5	0.5
EG	Bandgap voltage	eV	1.1	1.4
XTI	**IS** temperature exponent	-	0	
VTOTC	**VTO** temperature coefficient	V/°C	0	
BETATCE	**BETA** exponential temperature coefficient	%/°C	0	
TRG1	**RG** temperature coefficient (linear)	1/°C	0	0.001
TRD1	**RD** temperature coefficient (linear)	1/°C	0	0.001
TRS1	**RS** temperature coefficient (linear)	1/°C	0	0.001
KF	Flicker noise coefficient	-	0	-
AF	Flicker noise exponent	-	1	-

Parameters for Level 1				
Name	*Parameter*	*Units*	*Default*	*Typical*
ALPHA	Saturation voltage parameter	1/V	2.0	2.0
TAU	Conduction current delay time	s	0	
M	Gate *pn* grading coefficient	-	0.5	0.5

Parameters for Level 2				
Name	*Parameter*	*Units*	*Default*	*Typical*
ALPHA	Saturation voltage parameter	1/V	2.0	2.0
B	Doping tail extending parameter	1/V	0.3	0.3
TAU	Conduction current delay time	s	0	
M	Gate p-n grading coefficient	-	0.5	0.5
VDELTA	Capacitance transition voltage	V	0.2	0.2
VMAX	Capacitance limiting voltage	V	0.5	0.5

Parameters for Level3				
Name	*Parameter*	*Units*	*Default*	*Typical*
ALPHA	Saturation voltage parameter	1/V	2.0	2.0
GAMMA	Static feedback parameter	-	0	
DELTA	Output feedback parameter	1/AV	0	
Q	Power-law parameter	-	2	2
TAU	Conduction current delay time	s	0	
M	Gate *pn* grading coefficient	-	0.5	0.5
VDELTA	Capacitance transition voltage	V	0.2	0.2
VMAX	Capacitance limiting voltage	V	0.5	0.5

Parameters for Level 4				
Name	*Parameter*	*Units*	*Default*	*Typical*
ACGAM	Capacitance modulation	-	0	
DELTA	Output feedback parameter	1/AV	0	
Q	Power-law parameter	-	2	2
HFGAM	High-frequency V_{GD} feedback parameter	-	0	
HFG1	**HFGAM** modulation by V_{SG}	1/V	0	
HFG2	**HFGAM** modulation by V_{DG}	1/V	0	
HFETA	High-frequency V_{GS} feedback parameter	-	0	
HFE1	**HFETA** modulation by V_{GD}	1/V	0	

HFE2	**HFETA** modulation by V_{GS}	1/V	0	
LFGAM	Low-frequency feedback parameter	-	0	
LFG1	**LFGAM** modulation by V_{SG}	1/V	0	
LFG2	**LFGAM** modulation by V_{DG}	1/V	0	
MXI	Saturation knee-potential modulation	-	0	
MVST	Subthreshold modulation	1/V	0	
P	Linear-region power law exponent	-	2	2
TAUD	Relaxation time for thermal reduction	s	0	
TAUG	Relaxation time for GAM feedback	s	0	
VBD	Gate junction breakdown potential	V	1	5
VST	Subthreshold potential	V	0	0
XC	Capacitance pinch-off reduction factor	-	0	
XI	Saturation knee potential factor	-	1000	
Z	Knee transition parameter	-	0.5	
VMAX	Capacitance limiting voltage	V	0.5	0.5

C - Capacitor



General forms	*Restrictions*
C*name Pnode Nnode Value* [**IC**=*Init_cond*]	
C*name Pnode Nnode* **POLY** *c0 c1 c2* **...** [**IC**=*Init_cond*]	SPICE2
C*name Pnode Nnode* [*Value*] [*Model*] [**L**=*Length*] [**W**=*Width*] [**IC**=*Init_cond*]	SPICE3
C*name Pnode Nnode* [*Model*] *Value* [**IC**=*Init_cond*]	PSPICE
Examples	
CBL 3 0 10uF	
C3 3 5 100nF **IC**=1V	
CCOUP 7 12 47nF	
C12 4 5 CMODEL **L**=10u **W**=2u	SPICE3

This statement defines a capacitor with capacitance specified by *Value* (in farads) where *Pnode* and *Nnode* are the positive and negative nodes. An optional statement **IC**=*Init_cond* specifies the initial (time-zero) voltage (in volts) on the capacitance for transient analysis. This initial condition takes effect only when the **UIC** option is specified in the **.TRAN** statement.

A nonlinear capacitor can be defined using the **POLY** statement where *c0 c1 c2* ... are the coefficients of a polynomial describing the element value. The capacitance is expressed as a function of the voltage across the capacitor and is computed as

$$Value = c0+c1V+c2\,V^2+...$$

where V is the voltage across the capacitor. Although a nonlinear capacitor with keyword **POLY** was implemented in the original SPICE2 very few SPICE versions have this option implemented.

In SPICE3 and newer versions of SPICE the semiconductor capacitances can be declared. In this case the capacitance model is has to be specified in the **.MODEL** line:

.MODEL *Model* **C** [*model_papramters*] ...

1. Model parameters

Name	*Parameter*	*Units*	*Default*
CJ	Junction bottom capacitance	F/m^2	-
CJSW	Junction sidewall capacitance	F/m^2	-
DEFW	Default device width	m	1e-6
NARROW	Narrowing due to side etching	m	0.0

This more general model for the capacitor gives you the possibility of modeling capacitance values based on geometric and process information. If *Value* is given then information on geometry and process will be ignored. If *Model* is specified, the capacitance value is calculated based on information on process and geometry using the following formula:

$$\begin{aligned} Value &= \mathbf{CJ}\left(Length - \mathbf{NARROW}\right)\left(Width - \mathbf{NARROW}\right) \\ &\quad + 2\,\mathbf{CJSW}\left(Length + Width - 2\,\mathbf{NARROW}\right) \end{aligned} \tag{C-1}$$

If *Value* is not given, then *Model* and *Length* must be specified. If *Width* is not given, then the model default width **DEFW** will be used.

D - Diode



General forms	*Restrictions*
D*name Pnode Nnode Model* [*Rarea*] [**OFF**] [**IC**=*Vd*]	
D*name Pnode Nnode Model* [*Rarea*]	PSPICE
D*name Pnode Nnode Modell* [*Rarea*] [**OFF**] [**IC**=*Vd*] **[TEMP**=*T*]	SPICE3
Examples	
DREC 3 5 DMOD 0.2	
D1 7 12 SWITCH	
DBRIDGE 5 11 DIODEM 3	
DCLMP 9 12 DMOD2 2.0 **IC**=0.4V	
D3 3 4 DMOD **TEMP**=25	SPICE3

D*name* is the device name, and for the diode it must start with the letter **D**. *Pnode* and *Nnode* are the anode and cathode nodes, respectively. *Model* is the model name, and *Rarea* is the relative area factor. If *Rarea* is not specified, 1 is assumed. An optional parameter **IC**=*Vd* is used together with a **UIC** in a transient analysis. The keyword **OFF** indicates an optional starting condition for the dc analysis.

In SPICE3, the optional **TEMP** value is the temperature at which this device is to operate. It overrides the temperature specified in the **.OPTION** statement.

1. Diode model

.MODEL *Model_name* **D** [*Model parameters*]

2. Model parameters

Name	*Parameter*	*Units*	*Default*	*Typical*
IS	Saturation current for *Rarea* = 1	A	10^{-14}	10^{-14}
RS	Ohmic series resistance for *Rarea* = 1	Ω	0	3
N	Emission coefficient	-	1	1
TT	Transit time	s	0	10^{-9}
CJO	Zero-bias junction capacitance for *Rarea* = 1	F	0	$3\cdot10^{-12}$
VJ	Junction potential	V	1	0.8
M	Grading coefficient	-	0.5	0.5

EG	Energy gap	eV	1.11	1.11
XTI	Saturation current temperature exponent	-	3.0	3.0
KF	Flicker noise coefficient	-	0	-
AF	Flicker noise exponent	-	1	-
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5	-
BV	Reverse breakdown voltage	V	∞	80
IBV	Current at breakdown voltage	A	10^{-3}	$2{\cdot}10^{-3}$
TNOM	Temperature at which parameters were measured	°C	27	27
	PSPICE extensions			
IKF	Corner for high injection current roll-off for *Rarea* = 1	A	∞	0.1
TIKF	IKF temperature coefficient (linear)	1/°C	0	0
ISR	Recombination saturation current for *Rarea* = 1	A	0	10^{-8}
NR	Recombination emission coefficient	-	2	2
NBV	Reverse breakdown ideality factor	-	1	1
IBVL	Low-level reverse breakdown “knee” current for *Rarea* = 1	A	0	0
NBVL	Low-level reverse breakdown ideality factor	-	1	10^{-8}
TBV1	BV temperature coefficient (linear)	1/°C	0	0.003
TBV2	BV temperature coefficient (quadratic)	1/°C^2	0	0
TRS1	RS temperature coefficient (linear)	1/°C	0	0.002
TRS2	RS temperature coefficient (quadratic)	1/°C^2	0	0

Examples:

```
*Diode small power
.MODEL 1N3879 D( IS=1.6e-18 BV=50 IBV=10u M=0.27 CJO=100p RS=9m  TT=0.3u )

*Switching diode
.MODEL 1N4148 D(IS=0.1p RS=16 CJO=2p TT=12n BV=100 IBV=0.1p)

*Rectifier diode with 400 V breakdown voltage and 25 A
.MODEL 1N3494 D( IS=5E-14 BV=400 IBV=0.001 M=0.84 CJO=1.5NF  RS=3m  TT=8u)

*Germanium diode
.MODEL 1N5817 D(N=1.2 IS=20U RS=.08 EG=.69 XTI=2 CJO=200p BV=25 IBV=.01m
+ M=.523 VJ=2)

* Zener diode, 6.8 V
.MODEL 1N754   D(IS=1E-15 RS=.25  CJO=150p M=.55 VJ=.75 ISR=2n BV=6.8 IBV=20m )

* Variable-capacitance diode
.MODEL MV2201   D(IS=1p  CJO=15p M=.4261 VJ=.75  FC=.5  BV=25 IBv=10u)
```

E - Voltage-Controlled Voltage Source



General forms	*Restrictions*
E*name Pnode Nnode Pinp Ninp Gain*	
E*name Pnode Nnode* **POLY***(Dimensions) Pinp Ninp* **…** *Coef***…**	PSPICE
E*name Pnode Nnode* **VALUE** = {*Expression*}	PSPICE
E*name Pnode Nnode* **TABLE** {*Expression*} = (*Input*, *Output*) **…**	PSPICE
E*name Pnode Nnode* **LAPLACE**{*Expression*} = { *Transf_expression*}	PSPICE
E*name Pnode Nnode* **CHEBYSHEV**{*Expression*} ={**LB**|**HP**|**BP**|**BR**} + *Freq1 Freq2* ... *Atten1 Atten2* **…**	PSPICE
E*name Pnode Nnode* **FREQ** {*Expression*} = [[**MAG**|**DB**][**DEG**|**RAD**]|**R_I**] + (*Freq Magnitude Phase*) **…** [DELAY = *Delay*]	PSPICE

Examples	
E3 4 3 9 7 2.0	
EBUFF 3 7 4 6 100	
EAMP 4 6 POLY(1) 5 8 0 10	PSPICE
ENL 10 11 POLY(2) 7 0 8 3 0.0 7.5 0.1 2m	PSPICE
E13 5 0 VALUE = {12*SQRT(V(5,2))}	PSPICE
E3 4 5 TABLE {V(3) - V(1)}= (-2,4.2) (0,0) (3,5.6) (5,6)	PSPICE
E12 2 0 TABLE {V(3,8)} = (0,0) (25,1)	PSPICE
E3 8 0 LAPLACE {V(7)} = {1/(1+0.1*s)}	PSPICE
ELP 6 0 CHEBYSHEV {V(7)} = LP 1.6k 2.5K 0.5dB 60dB	PSPICE
ELP 7 0 FREQ {V(8)} = (0,0,0) (1kHz,0,0) (1.5kHz,-45,0) DELAY=1ms	PSPICE

The first of the general forms applies for linear sources, and this form is used in SPICE2 and SPICE3. Other forms are PSPICE extensions. *Pnode* is the positive node, and *Nnode* is the negative node. *Pinp* and *Ninp* are the positive and negative controlling nodes, respectively. *Gain* is the voltage gain. **POLY**(*Dimensions*) specifies the number of dimensions of the polynomial. The number of pairs of controlling nodes must be equal to the number of dimensions. A particular node may appear more than once, and the output and controlling nodes need not be different. The *Coef* parameter specifies coefficients of the polynomial.

The **VALUE**, **TABLE**, **LAPLACE**, **FREQ**, and **CHEBYSHEV** forms are part of the analog behavioral modeling feature of PSPICE. The **TABLE** form has a maximum size of 2048 input/output value pairs. If a **DELAY** value is specified, the simulator will modify the phases in the **FREQ** table to incorporate the specified delay value. This is useful for tables which the simulator identifies as being noncausal. When this occurs, the simulator provides a delay value necessary to make the table causal. The new syntax allows this value to be specified in subsequent simulation runs, without requiring the user to modify the table.

1. Function used in PSPICE

The functions can be used in the PSPICE expressions are listed in Table 5.2.

Table 5.2 Functions used in PSPICE

Function	*Description*	*Comments*
ABS(x)	$\lvert x \rvert$	
ACOS(x)	arccosine	$-1.0 \le x \le 1.0$
ARCTAN(x)	arctangent	result in radians
ASIN(x)	arcsine	result in radians
ATAN(x)	arctangent	result in radians
ATAN2(y, x)	$\tan^{-1}(y/x)$	result in radians
COS(x)	$\cos(x)$	x in radians
COSH(x)	hyperbolic cosine	
DDT(x)	time derivative of x	only for transient analysis
EXP(x)	e^x	
IF(t, x, y)	x if t = TRUE y if t = FALSE	t is a Boolean expression; x and y are either numerical values or expressions
IMG(x)	imaginary part of x	returns 0 for real numbers
LIM(x, min, max)	min if $x<min$ max if $x>max$ x otherwise	
LOG(x)	logarithm with base e	
LOG10(x)	logarithm with base 10	
M(x)	magnitude of x	the same as **ABS**(x)
MAX(x, y)	maximum of x and y	
MIN(x, y)	minimum of x and y	
P(x)	phase of x	returns 0 for real numbers
PWR(x, y)	$\lvert x \rvert^y$	can be replaced by {x**y}
PWRS(x, y)	$\lvert x \rvert^y$ if $x>0$ $-\lvert x \rvert^y$ if $x<0$	
R(x)	real part of x	
SGN(x)	1 if $x>0$ 0 if $x=0$ -1 if $x<0$	signum function
SIN(x)	$\sin(x)$	x in radians
SINH(x)	hyperbolic sine	
SQRT(x)	square root	
STD(x)	time integral of x	only for transient analysis
STP(x)	1 if $x>0$ 0 if $x\le 0$	step function
TABLE(x, x_1, x_2,…, x_n, y_n)	piecewise characteristics	
TAN(x)	tangent	x in radians
TANH(x)	hyperbolic tangent	

Chebyshev filters have two attenuation values, given in dB, which specify the pass-band ripple and the stop-band attenuation. They may be given in either order. Low-pass (**LP**) and high pass (**HP**) have two cutoff frequencies, specifying the pass-band and stop-band edges, while band- pass (**BP**) and band-reject (**BR**) filters must have four.

F - Current-Controlled Current Source



General forms	*Restrictions*
F*name Pnode Nnode Vname Gain*	
F*name Pnode Nnode* **POLY**(*dimensions*) *Vname* **…** *coef* **…**	PSPICE
Examples	
FSEN 7 8 VSENSE 180	
F5 5 9 VINP 50	
F7 4 8 VINP 100	
FA 4 7 POLY(1) VIN 0 1k	PSPICE
FNL 5 8 POLY(2) VCTRL1 VCTRL2 0.0 7.3 0.1 0.01	PSPICE

The first of the general form applies to the linear case. *Vname* is the name of a voltage source through which the controlling current flows. *Gain* is the current gain, *Pnode* is the positive node, and *Nnode* is the negative node. Current is directed from the positive node, through the source, to the negative node. The current through the controlling voltage source determines the output current. The direction of positive controlling current is from the positive node, through the source, to the negative node of *Vname*. The controlling source must be an independent voltage source (**V** device).

The second form is a PSPICE extension for the nonlinear case. **POLY***(dimensions)* specifies the number of dimensions of the polynomial. The number of controlling voltage sources must be equal to the number of dimensions .

G - Voltage-Controlled Current Source



<table>
<tr><th>General forms</th><th>Restrictions</th></tr>
<tr><td>Gname Pnode Nnode Pinp Ninp Gm</td><td></td></tr>
<tr><td>Gname Pnode Nnode POLY(Dimensions) Pinp Ninp … Coef …</td><td>PSPICE</td></tr>
<tr><td>Gname Pnode Nnode VALUE = {Expression}</td><td>PSPICE</td></tr>
<tr><td>Gname Pnode Nnode TABLE{Expression} =(Input, Output) …</td><td>PSPICE</td></tr>
<tr><td>Gname Pnode Nnode LAPLACE{Expression} ={ Transf_expression}</td><td>PSPICE</td></tr>
<tr><td>Gname Pnode Nnode CHEBYSHEV{Expression} ={LB|HP|BP|BR}
+ Freq1 Freq2 ... Atten1 Atten2 …</td><td>PSPICE</td></tr>
<tr><td>Gname Pnode Nnode FREQ {Expression} = [[MAG|DB][DEG|RAD]|R_I]
+ (Freq Magnitude Phase) … [DELAY = Delay]</td><td>PSPICE</td></tr>
</table>

Examples	
G1 2 3 11 13 10.0	
GBUFF 1 2 10 11 1.0	
GAMP 4 8 **POLY**(1) 3 7 0.0 100.0	PSPICE
GNL 3 8 **POLY**(2) 2 9 1 0 0.0 7.7 0.1 0.001	PSPICE
GSQRT 5 0 **VALUE** = {5V*SQRT(V(3,2))}	PSPICE
GT2 3 0 **TABLE** {V(4,8)} = (0,0) (50,1)	PSPICE
GRC 4 0 **LAPLACE** {V(5)} = {1/(1+.01*s)}	PSPICE
GLP 7 0 **CHEBYSHEV** {V(4)} = **LP** 1.6k 2.5k .2dB 40dB	PSPICE
GLP 6 0 **FREQ** {V(4)} = (0,0,0) (1kHz,0,0) (3kHz,-45,0) **DELAY**=2ms	PSPICE
GPSK 4 8 **VALUE** = {2mA*sin(6.28*10kHz*TIME+V(8))}	PSPICE
GT 4 9 **VALUE** = {20E-6***PWR**(V(1)*V(2),1.5)}	PSPICE
GLOSSY 3 6 **LAPLACE** {V(3)} = {exp(-sqrt(C*s*(R+L*s)))}	PSPICE

The first form applies for linear sources and is used in SPICE2 and SPICE3. Other forms are PSPICE extensions. *Pnode* is the positive node, and *Nnode* is the negative node. Positive current goes from the positive node through the source to the negative node. *Pinp* and *Ninp* are the positive and negative controlling nodes, respectively. *Gm* is the transconductance in A/V. **POLY**(*Dimensions*) specifies the number of dimensions of the polynomial. The number of pairs of controlling nodes must be equal to the number of dimensions. A particular node may appear more than once, and the output and controlling nodes need not be different. The *Coef* parameter specifies coefficients of the polynomial. Valid expressions for PSPICE are described with **E** source.

The **VALUE**, **TABLE**, **LAPLACE**, **FREQ**, and **CHEBYSHEV** forms are part of the analog behavioral modeling feature of PSPICE. The **TABLE** form has a maximum size of 2048 input/output value pairs. If a **DELAY** value is specified, the simulator will modify the phases in the **FREQ** table to incorporate the specified delay value. This is useful for tables which the simulator identifies as being noncausal. When this occurs, the simulator provides a delay value necessary to make the table causal. The new syntax allows this value to be specified in subsequent simulation runs, without requiring the user to modify the table.

Chebyshev filters have two attenuation values, given in dB, which specify the pass band ripple and the stop-band attenuation. They may be given in either order. Low-pass (**LP**) and high-pass (**HP**) have two cutoff frequencies, specifying the pass-band and stop-band edges, while band- pass (**BP**) and band-reject (**BR**) filters must have four.

H - Current-Controlled Voltage Source



General forms	*Restrictions*
H*name Pnode Nnode Vname Rm*	
H*name Pnode Nnode* **POLY**(*dimensions*) *Vname*… *Coef* …	PSPICE
Examples	
HSEN 3 7 V12 50.0	
HAMP 3 9 **POLY**(1) VINP 0.0 100.0	PSPICE
HNL 4 8 **POLY**(2) VCTRL1 VCTRL2 0.0 5.8 0.1 0.02	PSPICE

The first general form applies to the linear source. *Vname* is the name of a voltage source through which the controlling current flows, *Rm* is the transresistance, *Pnode* is the positive node, and *Nnode* is the negative node of the output voltage source. The current through the controlling voltage source determines the output current. The direction of positive controlling current is from the positive node, through the source, to the negative node of *Vname*. The controlling source must be an independent voltage source (**V** device).

The second form is the PSPICE extension for the nonlinear case. **POLY***(dimensions)* specifies the number of dimensions of the polynomial. The number of controlling voltage sources must be equal to the number of dimensions.

I - Independent Current Source



General forms	*Restrictions*
I*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] [*Signal_shape*]	
I*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] + [**STIMULUS** = *Name*] [*Signal_shape*]	PSPICE
I*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] [*Signal_shape*] + [**DISTOF1** *F1mag* [*F1phase*]] [**DISTOF2** *F2mag* [*F2phase*]]	SPICE3
Examples	
ISRC 4 8 **AC** 0.333 45.0	
INP 5 9 **DC** 1V **AC** 1mV 90	
IPULSE 3 0 **PULSE** (0 1 mA 5ns 1ns 1ns 100ns 200ns)	
I4 4 7 **DC** 5V **AC** 1mV **SIN**(0.01 0.001 1MEGHz)	
I9 6 9 **AC** 0.1 60 **SFFM** (0 1 100kHz 0.5 1kHz)	
IIN3 3 0 **AC** 1M **DISTOF1 DISTOF2** 0.001	SPICE3

A current source of positive value will force current out of the *Pnode* node, through the source, and into the *Nnode* node. *Value* is the dc and transient analysis value of the source. If the source value is time-invariant (e.g., a power supply), then the value may optionally be preceded by the letters **DC**.

Mag is the ac magnitude and *Phase* is the ac phase. The source is set to this value in the ac analysis. If *Mag* is omitted following the keyword **AC**, a value of unity is assumed. If *Phase* is omitted, a value of zero is assumed. If parameters other than source values are omitted or set to zero, the default values shown will be assumed. If a source is assigned a time-dependent value, the time-zero value is used for dc analysis.

The keyword STIMULUS is used in newer versions of PSPICE to call up custom signal shapes created with the stimulus editor. By specifying *Signal_shape*, a time-dependent waveform for transient analysis can be assigned. If a source is assigned a time-dependent value, the time-zero value is used for dc analysis. There are five independent source functions: pulse, exponential, sinusoidal, piecewise linear, and single-frequency FM. These five signal shapes are described in more detail in what follows.

1. Pulse waveforms

Form **PULSE** (I1 I2 TD TR TF PW PER)

Examples: I5 5 0 **PULSE**(-1mA 1mA 5ns 2ns 2ns 50ns 100ns)
I8 7 0 **PULSE**(0 5mA 5us 1us 1us 20us 50us)

Parameters	*Meaning*	*Default*	*Units*
I1	Initial value		A
I2	Pulsed value		A
TD	Delay time	0.0	s
TR	Rise time	*Tstep*	s
TF	Fall time	*Tstep*	s
PW	Pulse width	*Tstop*	s
PER	Period	*Tstop*	s

Parameters *Tstep* and *Tstop* are specified in the **.TRAN** statement.



2. Sine waves

Form **SIN**(IO IA FREQ TD DF PHASE)

Examples: I4 4 0 **SIN**(0 1mA 10kHz 10us 1k)
I7 3 9 **SIN**(0 5mA 1kHz)

parameters	*meaning*	*default*	*units*	*Restrictions*
IO	Offset		A	
IA	Amplitude		A	
FREQ	Frequency	1/*Tstop*	Hz	
TD	Delay	0.0	sec	
DF	Damping factor	0.0	1/sec	
PHASE	Phase	0.0	degree	PSPICE



$$\mathbf{I_S} = \mathbf{IA}\sin\left(2\pi\,\frac{\mathbf{PHASE}}{360}\right) \qquad \mathbf{(I\text{-}1)}$$

The shape of the waveform is described by the following expressions:

For time < **TD:**

$$\mathbf{IO} + \mathbf{IA}\sin\left[2\pi\left(\frac{\mathbf{PHASE}}{360}\right)\right] \qquad \mathbf{(I\text{-}2)}$$

For time > **TD:**

$$\mathbf{IO} + \mathbf{IA}\exp\left[-\left(\mathit{time}\,\text{-}\,\mathbf{TD}\right)\mathbf{THETA}\right]\sin\left[2\pi\left(\mathbf{FREQ}\left(\mathit{time}-\mathbf{TD}\right)+\frac{\mathbf{PHASE}}{360}\right)\right] \qquad \mathbf{(I\text{-}3)}$$

3. Exponential waveforms

Format **EXP**(I1 I2 TD1 TAU1 TD2 TAU2)

Examples: I5 5 0 **EXP**(-5mA 1mA 2ns 30ns 60ns 40ns)
I12 4 5 **EXP**(5mA 5us 10ns 15ns 18ns)

Parameters	*Meaning*	*Default*	*Units*
I1	Initial value		A
I2	Pulsed value		A
TD1	Rise delay time	0.0	s
TAU1	Rise time constant	*Tstep*	s
TD2	Fall delay time	TD1+*Tstep*	s
TAU2	Fall time constant	*Tstep*	s



The shape of the waveform is described by the following equations:

For *time* < **TD1**:

$$i(time) = 0 \qquad \textbf{(I-4)}$$

For **TD1** < *time* < **TD2**:

$$i(time) = \mathbf{I1} + (\mathbf{I2 - I1})\left[1 - \exp\left(-\frac{time - \mathbf{TD1}}{\mathbf{TAU1}}\right)\right] \qquad \textbf{(I-5)}$$

For *time* > **TD2**:

$$i(time) = \mathbf{I1} + (\mathbf{I2 - I1})\left[1 - \exp\left(-\frac{time - \mathbf{TD1}}{\mathbf{TAU1}}\right)\right] + (\mathbf{I1 - I2})\left[1 - \exp\left(-\frac{time - \mathbf{TD2}}{\mathbf{TAU2}}\right)\right] \qquad \textbf{(I-6)}$$

4. Piecewise linear waveforms

Format: **PWL**(T1 I1 [Tn In] …)

Example: ICLOCK 7 5 **PWL**(0 -7 10NS -7 11NS -3 17NS -3 18NS -7 50NS -7)

Parameters	*Meaning*	*Default*	*Units*
Tn	Time at corner	-	s
In	Current at corner	-	A

Each pair of values (**Tn**, **In**) specifies the value of the source **In** (in A) at *time*=**Tn**. The value of the source at intermediate values of time is determined by using linear interpolation of the input values.



5. FM waveforms

Form **SFFM** (IO IA FC MDI FS)

Examples: I7 6 0 **SFFM**(0 1mA 10kHz 0.5 1kHz)
I3 3 4 **SFFM**(5mA 1mA 200kHz 0.7 5kHz)

Parameters	*Meaning*	*Default*	*Units*
IO	Offset	-	A
IA	Amplitude	-	A
FC	Carrier frequency	*1/Tstop*	Hz
MDI	Modulation index	0	
FS	Signal frequency	*1/Tstop*	Hz



The **SFFM** (single-frequency frequency-modulated) waveform is described by the following equation:

$$i(time) = \mathbf{IO} + \mathbf{IA}\ \sin\left[2\pi\ \mathbf{FC}\,time + \mathbf{MDI}\sin\left(2\pi\,\mathbf{FS}\,time\right)\right] \qquad \textbf{(I-7)}$$

J - JFET



General forms	*Restrictions*
J*name Dnode Gnode Snode Model* [*Rarea*] [**OFF**] [**IC**=*Vds, Vgs*]	SPICE2
J*name Dnode Gnode Snode Model* [*Rarea*]	PSPICE
J*name Dnode Gnode Snode Model* [*Rarea*] [**OFF**] [**IC**=*Vds, Vgs*] + [**TEMP**=*T*]	SPICE3

Examples	
J4 3 5 9 JMOD	
J2 9 2 7 JM1 OFF	SPICE2/3

JFET is described by a statement that starts with the name of the JFET device **J***name*. This name must start with the letter **J**. Node numbers *Dnode*, *Gnode*, and *Snode* for drain, gate, and source follows the name. Next the model *Model* name is listed. Model parameters are specified in the **.MODEL** statement. Keywords **NJF** and **PJW** are used there for *n*-channel and *p*-channel, respectively. *Rarea* is the relative area factor. If *Rarea* is not specified, 1 is assumed. In SPICE2 and SPICE3, an optional parameter **IC**=*Vd* is used together with an **UIC** in a transient analysis. Keyword **OFF** indicates an optional starting condition for dc analysis. In SPICE3, different temperatures can be set for individual transistors using the keyword **TEMP**.

1. JFET models

.MODEL *Model_name* **NJF** [*Model parameters*]
.MODEL *Model_name* **PJF** [*Model parameters*]

2. Model parameters

Name	*Parameter*	*Units*	*Default*	*Typical*
VTO	Threshold voltage	V	-2.0	-2.0
BETA	Transconductance parameter	A/V^2	10^{-4}	10^{-4}
LAMBDA	Channel-length modulation parameter	1/V	0	0
RD	Drain resistance	Ω	0	20
RS	Source resistance	Ω	0	20
CGS	Zero-bias *G-S* junction capacitance	F	0	5 pF
CGD	Zero-bias *G-D* junction capacitance	F	0	5 pF
PB	Gate junction potential	V	1	0.8
IS	Gate junction saturation current	A	1.0^{-14}	1.0^{-15}
KF	Flicker noise coefficient	-	0	
AF	Flicker noise exponent	-	1	1
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5	0.5
TNOM	Parameter measurement temperature	°C	27	27
PSPICE extensions				
N	Gate *pn* emission coefficient	-	1	1
ISR	Gate *pn* recombination current parameter	A	0	
NR	Emission coefficient for **ISR**	-	2	2
ALPHA	Ionization coefficient	1/V	0	
VK	Ionization “knee” voltage	V	0	
M	Grading p-n coefficient	-	0.5	0.5
VTOTC	**VTO** temperature coefficient	V/°C	0	
BETACE	**BETA** exponential temperature coefficient	%/°C	0	
XTI	**IS** temperature coefficient	-	3	3

Examples:

```
* JFET p-type, analog switch; 40 V 50 mA, low Ron resistance
.MODEL J175 PJF (VTO=-5 BETA=3.6m  LAMBDA=7m  RD=15 RS=15 IS=3.5f CGS=12P
+ CGD=16P KF=5E-16)

* JFET n-type, analog switch; 40V 50 mA, low Ron resistance
.MODEL 1N4393 NJF (VTO=-1.50 BETA=4m LAMBDA=.035  RD=14 RS=15 IS=2E-15
+ CGS=7p CGD=9p  KF=1.5E-16
```

```
* JFET n-type, low noise, very high frequency
.MODEL 1N4416 NJF (VTO=-3.8 BETA=5.3m LAMBDA=.035 RD=35 RS=100 IS=5.E-15
+ CGS=6p CGD=3p  KF=3.246E-18)

*JFET n-type, general purpose  25 V, 10 mA
.MODEL 1N5457 NJF (VTO=-3 BETA=1.5m LAMBDA=5.16m  RD=40 RS=70 IS=5f
+CGS=15p CGD=4p KF=3E-17)

*JFET n-type, low-noise audio amplifier  30 V, 10 mA
.MODEL  BC264B   NJF  VTO=-1.8  BETA=1.2m  LAMBDA=18m RD= 0  RS= 0
+ IS = 0.3f  CGS= p CGD=2p  PB=0.77
```

K - Mutual Coupling



General forms	*Restrictions*
K*name Lname1 Lname2 Value*	
K*name Lname1 Lname2* **…** *Value* [*Model* [*size*]]	PSPICE
K*name Tname1 Tname2* **Cm**=*Cap_coupl* **Lm**=*Ind_coupl*	PSPICE

Examples	
K4 **L**5 **L**7 0.9	
KTUN **L**4 **L**7 0.95	
K5 **T**1 **T**7 **Cm**=10.0pF **Lm**=5mH	PSPICE
K12 **L**6 **L**8 **L**2 0.85	PSPICE
K6 **L**2 **L**3 0.99 KMOD	PSPICE

The mutual coupling statement describes a mutual inductive coupling between two inductors. *Lname1* and *Lname2* are the names of the two coupled inductors, and *Value* is the coupling coefficient K which must be greater than 0 and less than or equal to 1. Using the dot convention, place a dot on the first node of each inductor. The relation between the coupling coefficient K and the mutual inductance is given by:

$$M_{ij} = K\sqrt{L_i L_j} \qquad \textbf{(K-1)}$$

where L_i and L_j are the coupled pair of inductors, and M_{ij} is the mutual inductance between L_i and L_j.

A voltage induced in *ith* inductor L_i is given by:

$$v_i = L_i \frac{dI_i}{dt} + M_{ij} \frac{dI_j}{dt} + M_{ik} \frac{dI_k}{dt} + \cdots \qquad \textbf{(K-2)}$$

Newer versions of PSPICE include a model for inductive and capacitive coupling between two transmission lines using **Lm** and **Cm** coupling parameters. In the case of mutual coupling between transmission lines (names start with the letter **T**), two parameters can be specified. The **Cm** parameter describes the capacitive coupling in F/m, and the **Lm** parameter describes inductive coupling in H/m. PSPICE allows for declaration of coupling between more than two inductors.

In PSPICE the **CORE** model can be used to model lossy transformers; see: D. C. Jiles and D. L. Atherton, “Theory of Ferromagnetic Hysteresis”, *Journal of Magnetism and Magnetic Materials* **61**, 48 (1986).. The following **.MODEL** statement can accompany the mutual coupling declaration.

1. Core model

.MODEL *Model* **CORE** [.MODEL *name type* [*list_of_parameters*] …

2. Model parameters

Name	*Parameter*	*Units*	*Default*
AREA	Mean magnetic cross section	cm^2	0.1
PATH	Mean magnetic path length	cm	1
GAP	Effective air-gap length	cm	0
PACK	Pack (stacking) factor		1
MS	Magnetization saturation	A/m	10^{+6}
A	Thermal energy parameter	A/m	10^{+3}
C	Domain flexing parameter		0.2
K	Domain anisotropy parameter	A/m	500
ALPHA	Interdomain coupling parameter		10^{-3}
GAMMA	Domain damping parameter	1/s	

L - Inductor



General forms	*Restrictions*
L*name Pnode Nnode Value* [**IC**=*Init_cond*]	
L*name Pnode Nnode* **POLY** c0 c1 c2 **...** [**IC**=*Init_cond*]	SPICE2
Examples	
L3 7 9 15mH	
L5 4 6 3mH IC=10uA	
LS 1 9 50uH IC=2mA	

The inductor statement defines an inductor with inductance specified by *Value* (in H) where *Pnode* and *Nnode* are the positive and negative nodes. An optional statement **IC**=*Init_cond* specifies the initial (time-zero) current that flows from *Pnode*, through the inductor, to *Nnode*. This initial condition takes effect only when the **UIC** option is specified in the **.TRAN** statement.

In SPICE2, a nonlinear inductor can be defined using the **POLY** keyword where *c0 c1 c2* ... are the coefficients of a polynomial describing the element value. The inductance is expressed as a function of the current through the inductor and is computed as

$$Value = c0+c1I+c2\,I^2+\ldots \qquad \textbf{(L-1)}$$

Although the nonlinear inductor was originally implemented in SPICE2 with the keyword **POLY**, very few newer SPICE versions have this option implemented.

M - MOS Transistor



General forms	*Restrictions*
M*name Dnode Gnode Snode Bnode Model* [**L**=*Length*] [**W**=*Width*]	
+ [**AD**=*Darea*] [**AS**=*Sarea*] [**PD**=*Dperi*] [**PS**=*Speri*] [**NRD**=*Dsq*]	
+ [**NRS**=*Ssq*] [**OFF**] [**IC**=*Vds, Vgs, Vbs*]	
M*name Dnode Gnode Snode Bnode Model* [**L**=*Length*] [**W**=*Width*]	PSPICE
+ [**AD**=*Darea*] [**AS**=*Sarea*] [**PD**=*Dperi*] [**PS**=*Speri*] [**NRD**=*Dsq*]	
+ [**NRS**=*Ssq*] [**NRG**=*Gsq*] [**NRB**=*Bsq*] [**M**=*Value*]	
M*name Dnode Gnode Snode Bnode Model* [**L**=*Length*] [**W**=*Width*]	SPICE3
+ [**AD**=*Darea*] [**AS**=*Sarea*] [**PD**=*Dperi*] [**PS**=*Speri*] [**NRD**=*Dsq*]	
+ [**NRS**=*Ssq*] [**OFF**] [**IC**=*Vds, Vgs, Vbs*] [**TEMP**=*T*]	

Examples	
M1 3 7 9 0 PMOS **L**=5u **W**=20u	
M7 12 6 0 10 TYPEP	
M12 10 19 8 0 mosn w=5.6u l=67. 3u	
M4 4 6 2 0 MODN **L**=3u **W**=15u **AD**=200p **AS**=200pP **PD**=30u **PS**=40u	
M7 8 7 3 0 TYPEN	
MA 4 7 8 8 PNOM **L**=2.5u **W**=16u	
MB 3 6 9 20 PNOM **L**=2.5u **W**=10u **TEMP**=55	SPICE3
M4 4 5 2 0 NMOD **L**=5u **W**=40u **AD**=150p **AS**=150p **PD**=40u	PSPICE
+ **PS**=40u **NRD**=15 **NRS**=25 **NRG**=12	

M*name* is the device name, and in the case of the MOS transistor it must begin with the letter **M**. *Dnode*, *Gnode*, *Snode*, and *Bnode* are the drain, gate, source and bulk/substrate/well nodes, respectively. *Model* is the model name, and **L** and **W** are the channel length and width in meters. **AD** and **AS** are the drain and source diffusion areas in square meters. **PD** and **PS** are the perimeters of the drain and source lateral junctions in meters. **NRD** and **NRS** are the relative resistivities of the drain and source in number of squares. These parasitic resistances can be specified either by sheet resistance **RSH**, which is multiplied by **NRD** and **NRS**, or by **RD** and **RS** in the **.MODEL** definition. The calculation of resistance using the sheet resistance concept (resistance per square) is also explained in the resistor Section (Eq. R-1). Default values for **L**, **W**, **AD** and **AS** are **L** = 100 μm, **W** = 100 μm, **AD** = 0, and **AS** = 0. These default values can be changed with the **.OPTION** statement using **DEFL**, **DEFW**, **DEFAS**, and **DEFAD** keywords. Default values of **PD** and **PS** are 0.0, while default values of **NRD** and **NRS** are 1.0.

In SPICE2/3, the keyword **OFF** indicates an optional starting condition of the device for dc analysis. The optional initial value **IC** = *Vds, Vgs, Vbs* is used together with **UIC** in a transient analysis. In the case of the SPICE3, the optional **TEMP** value is the temperature at which this device operates and it overrides the temperature specified in the **.OPTION** statement.

In PSPICE, in addition to source and drain resistances, the user may specify the gate and bulk resistances using **NRG** and **NRB** parameters. **M** is a device multiplier which simulates the effect of multiple transistors connected in parallel.

1. MOS transistor models

.MODEL *Model_name* **NMOS** [*Model parameters*]
.MODEL *Model_name* **PMOS** [*Model parameters*]

A large number of MOS transistor models are used. These models are distinguished by the keyword LEVEL and a number. Some SPICE implementations (i.e. AIM-SPICE) have up to 20 different levels of MOS models. In this section three basic levels (1, 2, and 3), which are implemented in all SPICE versions, and the newer BSIM models, which are also becoming a standard, are described. Numbers in the brackets reefer to the reference list at the end of MOS section of Chapter 6.

LEVEL=1	Shichman-Hodges model [1] [8]	All SPICE implementations
LEVEL=2	Geometric-based analytical Meyer model [2] [8]	All SPICE implementations
LEVEL=3	Semi-empirical short channel Dang model [3] [8]	All SPICE implementations
LEVEL=4	BSIM1 (Berkeley Short Channel Igfet Model) [4] [9]	SPICE3 and new PSPICE
LEVEL=5	BSIM2 Jeng model [5] [9]	SPICE3
LEVEL=5	BSIM3 (version 1) [6] [9]	New PSPICE
LEVEL=6	BSIM3 (version 2) [6] [9]	New PSPICE
LEVEL=6	MOS6 Sakurai-Newton model [7]	SPICE3

2. Parameters of MOS transistor models

Common for all Levels				
Name	*Parameter description*	*Unit*	*Default*	*Typical*
LEVEL	Model index	-	1	
L	Default channel length (PSPICE only)	m	DEFL	100μ
W	Default channel width (PSPICE only)	m	DEFL	100μ
RD	Drain ohmic resistance	Ω	0	5
RS	Source ohmic resistance	Ω	0	5
RG	Gate ohmic resistance (PSPICE only)	Ω	0	5
RB	Bulk/substrate ohmic resistance (PSPICE only)	Ω	0	5
CBD	Zero-bias bulk-drain junction capacitance	F	0	20 fF
CBS	Zero-bias bulk-source junction capacitance	F	0	20 fF

IS	Bulk junction saturation current	A	10^{-14}	$3 \cdot 10^{-15}$
JS	Bulk junction saturation current per sq-meter of junction area	A/m^2	0	10^{-8}
JSSW	Bulk junction saturation current per length of sidewall area (PSPICE only)	A/m	0	10^{-12}
N	Bulk junction emission coefficient (PSPICE only)	-	1	1
PB	Bulk junction potential	V	0.8	0.85
PBSW	Bulk junction sidewall potential (PSPICE only)	V	**PB**	0.85
CGSO	Gate-source overlap capacitance per meter channel width	F/m	0	$3 \cdot 10^{-11}$
CGDO	Gate-drain overlap capacitance per meter channel width	F/m	0	$3 \cdot 10^{-11}$
CGBO	Gate-bulk overlap capacitance per meter channel length	F/m	0	$3 \cdot 10^{-10}$
RSH	Drain and source diffusion sheet resistance	Ω/	0	10
CJ	Zero-bias bulk junction bottom capacitance per square meter of junction area	F/m^2	0	$2 \cdot 10^{-4}$
CJSW	Zero-bias bulk junction sidewall capacitance per length of sidewall	F/m	0	10^{-8}
MJ	Bulk junction bottom grading coefficient	-	0.5	0.5
CJSW	Zero-bias bulk junction sidewall capacitance per meter of junction perimeter (PSPICE only)	F/m	0	10^{-9}
MJSW	Bulk junction sidewall grading coefficient (PSPICE only)	-	0.50 (Level 1) 0.33 (Level 2, 3)	
TT	Bulk junction transit time (PSPICE only)	s	0	10^{-8}
KF	Flicker noise coefficient	-	0	10^{-26}
AF	Flicker noise exponent	-	1.0	1.2
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5	0.5
TNOM	Nominal temperature which overwrites the value specified in **.OPTION** statement (SPICE3 only)	K	300	300

Level 1, 2, 3, and 6 (Sakurai-Newton)				
Name	*Parameter description*	*Unit*	*Default*	*Typical*
VTO	Zero-bias threshold voltage	V	0	1.0
KP	Transconductance parameter	A/V^2	$2{\cdot}10^{-5}$	$3{\cdot}10^{-5}$
GAMMA	Bulk threshold parameter	$V^{0.5}$	0	0.35
PHI	Surface potential	V	0.6	0.65
LAMBDA	Channel-length modulation parameter (level 1 and level 2 only)	1/V	0	0.02
TOX	Oxide thickness	m	10^{-7}	10^{-7}
NSUB	Substrate doping	cm^{-3}	0	$5{\cdot}10^{15}$
NSS	Surface state density	$cm\text{-}^{2}$	0	$2{\cdot}10^{10}$
NFS	Fast surface state density	$cm\text{-}^{2}$	0	10^{10}
TPG	Type of gate material (+1 for opposite to substrate, -1 for same as substrate, and 0 for Al gate)	-	1	1
XJ	Metallurgical junction depth	m		1u
LD	Lateral diffusion	m	0	0.7u
WD	Lateral diffusion width (PSPICE only)	m	0	0.5u
UO	Surface mobility	$cm^2/V\text{-}s$	600	700
UCRIT	Critical field for mobility degradation (level 2 only)	V/cm	10^4	10^4
UEXP	Critical field exponent in mobility degradation (level 2 only)	-	0	0.1
UTRA	Transverse field coefficient (mobility) (deleted for level 2)	-	0	0.3
VMAX	Maximum drift velocity of carriers	m/s	0	$3{\cdot}10^4$
NEFF	Total channel charge (fixed and mobile) coefficient (level 2 only)	-	1.0	3.0
XQC	Thin-oxide capacitance model flag and a fraction of channel charge attributed to drain (0-0.5)	-	1	0.4
DELTA	Width effect on threshold voltage	-	0	1.0
THETA	Mobility modulation (level 3 only)	1/V	0	0.1
ETA	Static feedback (level 3 only)	-	0	1.0
KAPP	Saturation field factor (level 3 only)	-	0.2	0.5

Transistor parameters may often be specified in different ways. For example, the reverse current can be specified either with the **IS** parameter (in A) or with **JS** (in A/m^2). The first choice is an absolute value, while the second choice is multiplied by **AD** and **AS** to give the reverse current at the drain and source junctions, respectively. The latter approach is preferred. The same is also true for the parameters **CBD**, **CBS**, and **CJ**. Parasitic resistances can be given with **RD** and **RS** (in Ω) or with **RSH** (in Ω/). **RSH** is multiplied by number of squares **NRD** and **NRS**.

Examples:

```
* NMOS transistor for 2um n-well MOSIS technology
.MODEL CMOSN NMOS LEVEL=2 PHI=0.7 TOX=40n XJ=0.2U TPG=1 VTO=0.8
+ DELTA=4.1 LD=0.3u KP=45u UO=550 UEXP=0.12  UCRIT=96k RSH=0.15 GAMMA=0.6
+ NSUB=7.3E+15 NFS=1.1E+11 VMAX=59k LAMBDA=0.03 CGDO=400p CGSO=400p
+ CGBO=350p CJ=0.1m MJ=0.6 CJSW=470p MJSW=0.3 PB=0.8

* PMOS transistor for 2um n-well MOSIS technology
.MODEL CMOSP PMOS LEVEL=2  PHI=0.7 TOX=40n XJ=0.2U TPG=-1 VTO=-0.9
+ DELTA=4.6 LD=0.35u KP=17u UO=205 UEXP=0.29 UCRIT=83k RSH=0.11 GAMMA=0.7
+ NSUB=9.5E+15 NFS=1.1E+11 VMAX=1MEG LAMBDA=0.046 CGDO=440p
+ CGSO=440p CGBO=390p CJ=0.3m MJ=0.6 CJSW=280p MJSW=0.4 PB=0.8
```

In the case of BSIM parameters for **LEVEL**=4, there are no default values, and all parameters must be specified. Also, some parameters, those marked with an asterisk “*” in the Table for Level 4, have channel length/width dependencies. For each of these parameters, two additional parameters should be specified. For example, if a parameter has the name **PNAM** then two additional parameters **LPNAM** and **WPNAM** should be specified. The actual parameter value is calculated using

$$\mathbf{PNAM} = \mathbf{PNAM} + \frac{\mathbf{LPNAM}}{\mathbf{L} - \mathbf{DL}} + \frac{\mathbf{WPNAM}}{\mathbf{W} - \mathbf{DW}}$$

where **L** and **W** are the channel length and width specified in the device line. Level 4 parameters were designed for automatic parameter extraction, and all model parameters should be copied from the device extractor rather than entered manually.

Level 4 - BSIM1			
Name	*Parameter description*	*Unit*	*L/W*
TOX	Gate oxide thickness	μm	
VFB	Flat band voltage	V	*
PHI	Surface inversion potential	V	*
K1	Body effect coefficient		*
K2	Drain/source depletion charge sharing coefficient	-	*

DL	Shortening of channel	μm	
DW	Narrowing of channel	μm	
N0	Zero-bias subthreshold slope coefficient	-	*
NB	Sensitivity of subthreshold slope to substrate bias	-	*
ND	Sensitivity of subthreshold slope to drain bias	-	*
VDD	Measurement bias range	V	
MUS	Mobility at zero substrate bias and at V_{DS} = **VDD**	$cm^2/V{\cdot}s$	
X2MS	Sensitivity of mobility to substrate bias at V_{DS} = **VDD**	$cm^2/V^2{\cdot}s$	*
X3MS	Sensitivity of mobility to drain bias at V_{DS} = **VDD**	$cm^2/V^2{\cdot}s$	*
MUZ	Zero-bias mobility	$cm^2/V{\cdot}s$	
X2MZ	Sensitivity of mobility to substrate bias at V_{DS} =0	$cm^2/V^2{\cdot}s$	*
U0	Zero-bias transverse-field mobility degradation coefficient	1/V	*
X2U0	Sensitivity of transverse field mobility degradation effect to substrate bias	$1/V^2$	*
U1	Zero-bias velocity saturation coefficient	μm/V	*
X2U1	Sensitivity of velocity saturation effect to substrate bias	$\mu m/V^2$	*
X3U1	Sensitivity of velocity saturation effect on drain bias at V_{DS} = **VDD**	$\mu m/V^2$	*
WDF	Source-drain junction default width	m	
DELL	Source-drain junction length reduction	m	
TEMP	Temperature at which parameters are measured	^oC	
ETA	Zero-bias drain-induced barrier-lowering coefficient	-	*
X2E	Sensitivity of drain-induced barrier-lowering effect to substrate bias	1/V	*
X3E	Sensitivity of drain-induced barrier-lowering effect to drain bias at V_{DS} = **VDD**	1/V	*
XPART	Gate-oxide capacitance charge model flag. **XPART** = 0 selects a 40/60 drain/source partition of the gate charge in saturation, while **XPART** = 1 selects a 0/100 drain/source charge partition.	-	

O - Lossy Transmission Lines (LTRA) SPICE3 only



General forms	*Restrictions*
O*name n1 n2 n3 n4 Model* (In PSPICE, lossy transmission lines use names which start with the letter **T**)	SPICE3
Examples	
O5 2 0 7 0 lmod	SPICE3
O15 7 3 12 3 connection	SPICE3

1. LTRA model

.MODEL *Model* **LTRA** [*list_of_parameters*]

2. Model parameters SPICE3 only

Name	*Parameter*	*Units*	*Default*	*Restrictions*
R	Resistance/length.	Ω/m	0.0	SPICE3
L	Inductance/length.	H/m	0.0	SPICE3
C	Capacitance/length.	F/m	0.0	SPICE3
G	Conductance/length.	1/Ω·m	0.0	SPICE3
LEN	Length of line.	m	-	SPICE3
REL	Breakpoint control.	-	1	SPICE3
ABS	Breakpoint control.	-	1	SPICE3
NOSTEPLIMIT	Don't limit time step to less than line delay. This flag will remove the default restrictions of limiting the time-step to less than the line delay in the *RLC* case.	Flag	Not set	SPICE3
NOCONTROL	Don't do complex timestep control. This flag prevents the default limitation on the time-step based on convolution error criteria in the *RLC* and *RC* cases. This speeds up the simulation, but may in some cases reduce the accuracy.	Flag	Not set	SPICE3

LININTERP	Use linear interpolation. When this flag is set, linear interpolation is used instead of the default quadratic interpolation for calculating delayed signals.	Flag	Not set	SPICE3
MIXEDINTERP	When this flag is set, SPICE uses a metric for judging whether quadratic interpolation is applicable, and if not so, it uses linear interpolation. Otherwise the default quadratic interpolation is used.	Flag	Not set	SPICE3
COMPACTREL	Special **RELTOL** for history compacting	-	**RELTOL**	SPICE3
COMPACTABS	Special **ABSTOL** for history compacting	-	**ABSTOL**	SPICE3
TRUNCNR	Use Newton-Raphson method for time step control. This flag initiates Newton-Raphson iterations to determine an appropriate time step in the time step control routines. The default is a trial-and-error procedure which cuts the previous time step in half.	Flag	Not set	SPICE3
TRUNCDONTCUT	Don't limit time step to keep impulse-response errors low. This flag removes the default cutting of the time step to limit errors in the actual calculation of impulse-response related quantities.	Flag	Not set	SPICE3

Example:

```
* coaxial cable with Z0=50 ohms and 100 pF/m 100 m long
.MODEL LOSSY LTRA(R=2.5 G=0 L=250n C=100P LEN=100)
```

LTRA uses a two-port convolution model for lossy transmission lines. *n1* and *n2* are the nodes at port 1, and *n3* and *n4* are the nodes at port 2. It is worth mentioning that a lossy transmission line with zero loss may be more accurate than the lossless transmission line. The length **LEN** of the line must be specified.

The following types of lines are implemented :

- *RLC* - uniform transmission line with series loss only
- *RC* - uniform *RC* line
- *LC* - lossless transmission line
- *RG* - distributed series resistance and parallel conductance only

Other line structures may lead to erroneous results

Q - Bipolar Transistor



General forms	*Restrictions*
Q*name Cnode Bnode Enode* [*Snode*] *Model* [*Rarea*] [**OFF**] [**IC**=*Vbe*, *Vce*]	
Q*name Cnode Bnode Enode* [*Snode*] *Model* [*Rarea*]	PSPICE
Q*name Cnode Bnode Enode* [*Snode*] *Model* [*Rarea*] [**OFF**] [**IC**=*Vbe*, *Vce*] +[**TEMP**=*T*]	SPICE3
Examples	
Q3 4 5 6 0 QMOD **OFF**	
Q7 4 6 0 QMOD2 **IC**=0.7,10	
Q12 4 7 9 QPNP	
Q17 7 3 9 QNPN **IC**=0.6,5.0 **TEMP**=55	SPICE3

Q*name* is the device name for a bipolar junction transistor and it must begin with the letter **Q**. *Cnode*, *Bnode*, *Enode*, and *Snode* are the collector, base, emitter and substrate nodes, respectively. For the case of **NPN** and **PNP** transistors, the substrate node is associated with the collector-substrate diode. For the case of **LPNP** (lateral *pnp* transistor) the substrate is associated with the base-substrate diode. If *Snode* is not given, ground is assumed. *Model* is the model name and *Rarea* is the relative area. If *Rarea* is omitted, the default *Rarea*=1.0 is assumed. The keyword **OFF** indicates an optional starting condition of the device for dc analysis. The optional initial value **IC**=*Vbe*, *Vce* is used together with **UIC** in transient analysis. In the case of the SPICE3, the optional **TEMP** value is the temperature at which this device operates, and it overrides the temperature specified In the **.OPTION** statement.

1. Bipolar transistor models

.MODEL *Model_name* **NPN** [*Model parameters*]
.MODEL *Model_name* **PNP** [*Model parameters*]
.MODEL *Model_name* **LPNP** [*Model parameters*] PSPICE only

2. Parameters of bipolar transistor model (modified Gummel-Poon model)

Name	*Parameter description*	*Unit*	*Default*	*Typical*
IS	Saturation current for *Rarea*=1	A	10^{-16}	10^{-15}
ISE	*B-E* leakage saturation current for *Rarea*=1	A	0	10^{-12}
ICS	*B-C* leakage saturation current for *Rarea*=1	A	0	10^{-12}
BF	Forward current gain	-	100	100

BR	Reverse current gain	-	1	0.1
NF	Forward current emission coefficient	-	1.0	1.2
NR	Reverse current emission coefficient	-	1.0	1.3
NE	B-E leakage emission coefficient	-	1.5	1.4
NC	B-C leakage emission coefficient	-	1.5	1.4
VAF	Forward Early voltage	V	∞	100
VAR	Reverse Early voltage	V	∞	50
IKF	$ß_F$ high current roll-off corner	A	∞	0.05
IKR	$ß_R$ high current roll-off corner	A	∞	0.01
IRB	Current where base resistance falls by half for *Rarea*=1	A	∞	0.1
RB	Zero-bias base resistance	Ω	0	100
RBM	Minimum base resistance	Ω	**RB**	10
RE	Emitter series resistance for *Rarea*=1	Ω	0	1
RC	Collector series resistance for *Rarea*=1	Ω	0	50
CJE	*B-E* zero-bias depletion capacitance	F	0	10^{-12}
CJC	*B-C* zero-bias depletion capacitance	F	0	10^{-12}
CJS	Zero-bias collector-substrate capacitance	F	0	10^{-12}
VJE	*B-E* built-in potential	V	0.75	0.8
VJC	*B-C* built-in potential	V	0.75	0.7
VJS	Substrate junction built-in potential	V	0.75	0.7
MJE	*B-E* junction exponential factor	-	0.33	0.33
MJC	*B-C* junction exponential factor	-	0.33	0.5
MJS	Substrate junction exponential factor	-	0	0.5
XCJC	Fraction of *B-C* capacitance connected to internal base node (see Fig. 6)	-	0	0.5
TF	Forward transit time	s	0	10^{-10}
TR	Reverse transit time	s	0	10^{-8}
XTF	Coefficient for bias dependence of τ_F	-	0	-
VTF	Voltage for t_F dependence on V_{BC}	V	∞	-
ITF	Current where $t_F = f(I_C,V_{BC})$ starts	A	0	-
PTF	Excess phase at $freq = 1/(2pt_F)$ Hz	deg	0	-
XTB	Forward and reverse beta temperature exponent		0	-

EG	Energy gap	eV	1.11	1.1
XTI	Temperature exponent for effect on I_s	-	3	3.5
KF	Flicker noise coefficient	-	0	
AF	Flicker noise exponent	-	1	
FC	Coefficient for the forward biased depletion capacitance formula	-	0.5	0.5
SPICE3 extension				
TNOM	Nominal temperature which overrides the value specified in **.OPTION** statement	K	300	300
PSPICE extensions				
NK	High-current roll-off coefficient	-	0.5	0.5
ISS	Substrate saturation current for *Rarea*=1	A	0	10^{-15}
NS	Substrate emission coefficient	-	1	1
QCO	Epitaxial layer charge factor for *Rarea*=1	C	0	
RCO	Epitaxial region resistance for *Rarea*=1	Ω	0	100
VO	Carrier mobility knee voltage	V	10	20
GAMMA	Epitaxial layer doping factor		10^{-11}	10^{-11}
TRE1	**RE** temperature coefficient (linear)	1/°C	0	0.001
TRE2	**RE** temperature coefficient (quadratic)	1/°C^2	0	0
TRB1	**RB** temperature coefficient (linear)	1/°C	0	0.002
TRB2	**RB** temperature coefficient (quadratic)	1/°C^2	0	0
TRM1	**RBM** temperature coefficient (linear)	1/°C	0	0.002
TRM2	**RBM** temperature coefficient (quadratic)	1/°C^2	0	0
TRC1	**RC** temperature coefficient (linear)	1/°C	0	0.003
TRC2	**RC** temperature coefficient (quadratic)	1/°C^2	0	0

Examples:

```
* small power general purpose npn transistor
.MODEL 2N2222 NPN (IS=15.2f NF=1 BF=105 VAF=98.5  IKF=.5 ISE=8.2p NE=2
+ BR=4 NR=1 VAR=20 IKR=.225 RE=.373 RB=1.49 RC=.149 XTB=1.5 CJE=35.5p
+ CJC=12.2P TF=0.5n TR=85n)

* small power general purpose pnp transistor 40 V, 200 mA
.MODEL 2N2904 PNP (IS=0.3n  NF=1 BF=100 VAF=120  IKF=.14 ISE=46.1p NE=2  BR=4
+ NR=1 VAR=20 IKR=.2 RE=.5 RB=2 RC=.2 XTB=1.5 CJE=15p CJC=20p TF=600p TR=60n)
```

```
* small power general purpose npn transistor 40 V, 200 mA
.MODEL 2N3903 NPN (IS=1F NF=1 BF=400 VAF=120 IKF=70m  ISE=3P NE=2 BR=4 NR=1
+ VAR=20 RE=21 RB=8 RC=1 XTB=1.5 CJE=8p CJC=5p TF=600p TR=0.3u)

*small power general purpose npn transistor 45 V, 200mA
.MODEL BC107A NPN (IS=10f NF=1 BF=300  VAF=120 IKF=0.05 ISE=5p NE=2 BR=4
+  NR=1 VAR=30  XTB=1.5 RE=1 RB=3 RC=0.3 CJE=15p CJC=5p TF=0.5n TR=60n)

*small power general purpose pnp transistor 45 V, 200 mA
.MODEL BC177A PNP (IS=0.1f ISE=0.5f NF=1 NE=1.4 BF=300 BR=13 IKF=.1 IKR=.01
+ ISC=0.1f NC=1.1 NR=1 RB=.2 RE=.4 RC=1 VAR=10 VAF=90 CJE=16p TF=.5n
+ CJC=10p  TR=70n MJC=.4 VJC=.6)

*small power germanium transistor 25 V, 100 mA
.MODEL 2N2955 PNP (IS=1.25n NF=1 BF=80 VAF=90 IKF=60m  ISE=5n NE=2 BR=4 NR=1
+ VAR=14 IKR=90m RE=2 RB=10 RC=1 XTB=1.5 CJE=30p CJC=9p TF=0.4n TR=20n)

* power npn transistor 15 A, 100V, 100 W
.MODEL 2N3055 NPN(IS=5p NF=1 BF=100 VAF=100 IKF=.25  ISE=30p ISC=5n RB=3
+ IRB=1m RBM=.4 NE=1.5 RC=.04 BR=3 MJC=.4 VJE=1 MJE=.45 XTB=1 CJE=600p
+ TF=80n CJC=200p TR=2u PTF=120 XTF=1 ITF=3)

* power npn transistor 4 A, 40 V
.MODEL 2N5190 NPN (IS=5p NF=1 BF=150 VAF=120 IKF=0.3 ISE=0.7n NE=2 BR=4
+ NR=1 VAR=20  XTB=2.5 RE=0.2 RB=12 RBM=1.2 IRB=0.5m RC=0.07  CJE=0.3n
+ CJC=0.3n TF=45n TR=1u  PTF=120 XTF=1 ITF=3.5  ISC=5n MJC=0.2 VJC=1.2
+ MJE=0.3 VJE=0.5)

* power pnp transistor 4 A, 40 V
.MODEL 2N5193 PNP (IS=0.4p NF=1 BF=100 VAF=100 IKF=0.3 ISE=0.3n ISC=7n NE=2
+ BR=4 NR=1 VAR=20  XTB=1.4 RE=0.15 RB=15 RBM=1.5 IRB=0.3m RC=0.06 CJE=0.3n
+ CJC=0.5n VJC=1.25 MJE=0.3 VJE=0.65)
```

R - Resistor



General forms	*Restrictions*
R*name Pnode Nnode Value* [**TC**=*TC1* [*TC2*]]	
R*name Pnode Nnode* [*Value*] [*Model*] [**L**=*Length*] [**W**=*Width*] + [**TEMP**=*T*]	SPICE3
R*name Pnode Nnode* [*Model*] [*Value*] [**TC**=*TC1* [*TC2*]]	PSPICE
Examples	
R1 3 7 1k	
RC 7 9 10k **TC**=0.02,0.0015	
RL 3 9 3.7k	
RLOAD 2 12 RMODEL **L** = 48um **W** = 3um	SPICE3
R12 2 9 RMOD 800k **TC** = 0.01, 0.0015	PSPICE

The resistor statement consists of a name which must start with the letter **R**, node names *Pnode* and *Nnode*, and a value of resistance specified by *Value* (in ohms). An ptional **TC**=*TC1* [*TC2*]] specifies the temperature dependence of resistance, where *TC1* and *TC2* are linear and quadratic temperature coefficients, respectively.

The basic resistor statement has many extensions that are implementation-dependent. Semiconductor resistors are implemented in SPICE3. This extension models temperature effects and calculates the resistance based on geometry and processing information. If *Value* is given, then the *Value* defines the resistance, and information on geometry and processing is ignored. If *Model* is specified, the resistance value is calculated based on information about the process and geometry in the model statement:

$$R = \mathbf{RSH}\,\frac{Length - \mathbf{NARROW}}{Width - \mathbf{NARROW}} \qquad \textbf{(R-1)}$$

If *Value* is not given, *Model* and *Length* must be specified. If *Width* is not given, it will be given the default value. The optional TEMP value is the temperature at which this device operates. It overrides the default temperature specified in the **.OPTION** statement. The temperature dependence of the resistance is calculated using

$$R(T) = R(\mathbf{TNOM})\left[1 + TC1\,(T - \mathrm{TNOM}) + TC2\,(T - \mathrm{TNOM})^2\right] \qquad \textbf{(R-2)}$$

The resistor model contains process-related parameters, and the resistance value is a function of the temperature.

1. Resistor model

.MODEL *Model* **R** [*list_of_parameters*]

2. Model parameters

Name	*Parameter*	*Units*	*Default*	*Restrictions*
TC1	First-order temperature coefficient	1/°C	0.0	
TC2	Second order temperature coefficient	$1/°C^2$	0.0	
RSH	Sheet resistance	Ω/	-	SPICE3
DEFW	Default width	m	1e-6	SPICE3
NARROW	Narrowing due to side etching	m	0.0	SPICE3
TNOM	Parameter measurement temperature	°C	27	SPICE3
R	Resistance multiplier	-	1	PSPICE
TCE	Exponential temperature coefficient	%/°C	0	PSPICE

PSPICE uses the **TCE** parameter to calculate the temperature dependence of resistance:

$$R(T) = R(T_{nom})\, 1.01^{\mathbf{TCE}(T - T_{nom})} \qquad \textbf{(R-3)}$$

S - Voltage Controlled Switch



General forms	*Restrictions*
S*name Pnode Nnode Pinp Ninp Model*	PSPICE
S*name Pnode Nnode Pinp Ninp Model* [**ON**] [**OFF**]	SPICE3
Examples	
S7 4 7 2 9 SW1	
SW5 2 8 3 0 Smodel	
S4 4 9 6 2 SW2 **OFF**	SPICE3
S3 3 7 2 9 sw1 **ON**	SPICE3
s3 3 6 8 0 SMOD **off**	SPICE3

The voltage-controlled switch statement begins with the letter **S**. *Pnode* and *Nnode* represent the connections to the switch terminals. *Pinp* and *Ninp* are positive end negative controlling nodes, respectively. The model name, *Model*, is mandatory, while the initial conditions are optional. The controlling voltages are defined the same way as in voltage-controlled voltage (letter **E**) and current (letter **G**) sources. In SPICE3, the optional parameter **ON** or **OFF** specifies the switch state for the dc operating point.

The switch model allows an almost ideal switch to be described in SPICE. The switch is not quite ideal, in that the resistance cannot change from 0 to infinity, but must always have a finite and nonzero positive value. By proper selection of the on and off resistances, they can be effectively zero and infinity in comparison to other circuit elements.

The switch can have a hysteresis described by the **VH** parameter. For example, the voltage-controlled switch will be in the on state, with a resistance **RON**, at **VT**+**VH**. The switch will be in the off state, with a resistance **ROFF**, at **VT-VH**.

1. SW model

.MODEL *Model* **SW** [*list_of_parameters*] SPICE3

2. SW model parameters

Name	*Parameter*	*Default*	*Units*	*Restrictions*
VT	Threshold voltage	0.0	V	SPICE3
VH	Hysteresis voltage	0.0	V	SPICE3
RON	On resistance	1.0	Ω	
ROFF	Off resistance	1/**GMIN**	Ω	SPICE3

Examples:
.MODEL SMOD **SW** **RON**=1m **ROFF**=20k VT=3V
.MODEL SMOD **SW** **VT**=4V **VH**=1V
.MODEL SMOD **SW** **RON**=10 **ROFF**=10MEG **VT**=2V

3. VSWITCH model

.MODEL *Model* **VSWITCH** [*list_of_parameters*] PSPICE

4. VSWITCH model parameters

Name	*Parameter*	*Default*	*Units*	*Restrictions*
VON	Threshold current for on state	10^{-3}	V	PSPICE
VOFF	Threshold current for off state	0.0	V	PSPICE
RON	On resistance	1.0	Ω	
ROFF	Off resistance	10^{6}	Ω	PSPICE

The use of an ideal element that is highly non-linear, such as a switch, can cause large discontinuities to occur in the circuit node voltages. The rapid voltage change associated with a switch changing state can cause numerical round off or tolerance problems leading to erroneous results or time step difficulties. You can improve the situation by taking the following steps. Set the switch impedances only high and low enough to be negligible with respect to other elements in the circuit. Using switch impedances that are close to “ideal” under all circumstances will aggravate the discontinuity problem. When modeling real devices such as MOSFETS, the on resistance should be adjusted to a realistic level depending on the size of the device being modeled.

If a wide range of **ON** to **OFF** resistance must be used (**ROFF/RON** $>10^{+12}$), then the tolerance on errors allowed during transient analysis should be decreased by specifying the **.OPTIONS TRTOL** parameter to be less than the default value of 7.0. When switches are placed around capacitors, the **.OPTIONS CHGTOL** parameters should also be reduced. Suggested values for these two options are 1.0 and 1E-16, respectively. These changes inform SPICE to be more careful near the switch points so that no errors are made due to the rapid change in the circuit response.

T - Transmission Lines



General forms	*Restrictions*
T*name n1 n2 n3 n4* **Z0**=*Value* [**TD**=*Delay*] [**F**=*Freq* [**NL**=*Wavelength*]] +[**IC**=V1, I1, V2, I2]	SPICE2/3
T*name n1 n2 n3 n4 Model* [**IC**=V1, I1, V2, I2]	PSPICE ideal
T*name n1 n2 n3 n4* **LEN**=*Electrical_length* **R**=*Resistance_per_length* + **L**=*Inductance_per_length* **G**=*Conductance_per_length* + **C**=*Capacitance_per_length*	PSPICE lossy
(In SPICE3, lossy transmission lines use names which start with the letter **O**)	SPICE3

Examples	
T1 1 0 70 **Z0**=50 **TD**=10ns 12 17 1k	SPICE2/3
T1 2 0 3 0 **Z0**=75 **TD**=10ns **IC**=1V,1mA,1V,1mA	SPICE2/3
T1 2 6 4 9 **Z0**=200 **TD**=115ns	SPICE2/3
T2 3 6 2 8 **Z0**=75 **F**=1MEG	SPICE2/3
T3 3 5 7 9 **Z0**=50 **F**=4.5MEG **N**L=0.5	SPICE2/3
T1 2 5 8 2 **LEN**=1 **R**=.25 L=.5u **G**=10u **C**=50p	PSPICE
T3 2 7 5 9 TMOD 1	PSPICE

n1 and *n2* are the nodes at port 1; *n3* and *n4* are the nodes at port 2. For the ideal case, **Z0** is the characteristic impedance. The transmission line's length can be specified either by **TD**, a delay in seconds, or by **F** and **NL**, a frequency and a relative wavelength at **F**. **NL** defaults to 0.25 (**F** is then the quarter-wave frequency). Although **TD** and **F** are both shown as optional, one of the two must be specified.

Note that this element models only one propagating mode. If all four nodes are distinct in the actual circuit, then two modes may be excited. To simulate such a situation, two transmission-line elements are required. The (optional) initial condition specification consists of the voltage and current at each of the transmission line ports. Note that the initial conditions (if any) apply 'only' if the **UIC** option is specified on the **.TRAN** line. One should be aware that SPICE will use a transient time step which does not exceed 1/2 the minimum transmission line delay. Therefore, very short transmission lines (compared with the analysis time frame) will cause long run times.

For the case of a lossy line, **LEN** is the electrical length. **R, L**, **G**, and **C** are the per unit length values of resistance, inductance, conductance, and capacitance, respectively. The lossy line model is similar to that shown for the ideal case, except that the delayed voltage and current values include terms that vary with frequency. These terms are computed in transient analysis using an impulse response convolution method, and the internal time step is limited by the time resolution required to accurately model the frequency characteristics of the line. As with ideal lines, short lossy lines will cause long run times.

For the case of a line that uses a model, the electrical length is given after the model name. All of the transmission line parameters from either the ideal, or lossy parameter set can be expressions. In addition, **R** and **G** can be general Laplace expressions. This option allows the user to model frequency-dependent effects, such as skin effect and dielectric loss. However, this adds to the computation time for transient analysis, since the impulse responses must be obtained by an inverse FFT.

The simulator uses a distributed model to represent the properties of a lossy transmission line, and the line resistance, inductance, conductance, and capacitance are all continuously apportioned along the line's length. A common approach to simulating lossy lines is to model these characteristics using discrete passive elements to represent small sections of the line. This is the lumped model approach, which involves connecting a set of many small subcircuits in series.

An additional PSPICE extension allows systems of coupled transmission lines to be simulated. Transmission line coupling is specified using the **K** mutual coupling device. This is done in much the same way that coupling is specified for inductors.

The distributed model used in the simulation process frees you from having to determine how many lumps are sufficient, and eliminates spurious oscillations. It also allows lossy lines to be simulated in a fraction of the time necessary when using the lumped approach, for the same accuracy.

1. TRN model

.MODEL *Model* **TRN** [*list_of_parameters*]

2. Model parameters

PSPICE only

Name	*Parameter*	*Units*	*Default*	*Restrictions*
	Ideal transmission line			
ZO	Characteristic impedance	Ω		PSPICE
TD	Transmission delay	s		PSPICE
NL	Relative wavelength		0.25	PSPICE
F	Frequency for **NL**	Hz		PSPICE
	Lossy transmission line			
LEN	Electrical length	Any unit		PSPICE
R	Resistance per **LEN** units	Ω/unit		PSPICE
L	Inductance per **LEN** unit	H/unit		PSPICE
C	Capacitance per **LEN** unit	F/unit		PSPICE
G	Conductance per **LEN** unit	1/Ω·unit		PSPICE

U - Uniform Distributed RC Lines (URC) SPICE3 only



General form	*Restrictions*
U*name n1 n2 n3 Model* **L**=*Length* [**N**=*Lumps*]	SPICE3
Examples	
U7 4 2 8 UMOD **L**=500um	SPICE3
URC 3 6 9 URC **L**=1mil **N**=12	SPICE3

n1 and *n2* are the two nodes of the *RC* line itself, while *n3* is the capacitive node. *Model* is the name of the model, *Length* is the length of the line in meters and *Lumps*, if given, is the number of segments to use in modeling the RC line.

1. URC model

.MODEL *Model* **URC** [*list_of_parameters*]

2. Model parameters SPICE3 only

Name	*Parameter*	*Units*	*Default*	*Restrictions*
K	Propagation constant	-	2.0	SPICE3
FMAX	Maximum frequency	Hz	10^9	SPICE3
RPERL	Resistance per unit length	Ω/m	1000	SPICE3
CPERL	Capacitance per unit length	F/m	10^{-15}	SPICE3
ISPERL	Saturation current per unit length	A/m	0	SPICE3
RSPERL	Diode resistance per unit length	Ω /m	0	SPICE3

The model is accomplished by a subcircuit expansion of the **URC** line into a network of lumped *RC* segments with internally generated nodes. The *RC* segments are in a geometric progression, increasing toward the middle of the **URC** line, with **K** as a proportionality constant. **N** is the number of lumped segments used. If not specified on the **URC** line, **N** is determined by the following expression:

$$\mathbf{N} = \frac{\log\left[2\pi RCF_{\max}\left(\frac{\mathbf{K}-1}{\mathbf{K}}\right)^2\right]}{\log\mathbf{K}} \qquad \textbf{(U-1)}$$

The URC line is made up strictly of resistor and capacitor segments unless the **ISPERL** parameter is given a nonzero value, in which case the capacitors are replaced with reverse-biased diodes with a zero-bias junction capacitance equivalent to the capacitance replaced, and with a saturation current of **ISPERL** amps per meter of transmission line. An optional series resistance equivalent to **RSPERL** ohms per meter can be included.

V - Independent Voltage Source



General forms	*Restrictions*
V*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] [*Signal_shape*]	
V*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] + [**STIMULUS** = *Name*] [*Signal_shape*]	PSPICE
V*name Pnode Nnode* [[**DC**] *Value*] [[**AC**] *Mag* [*Phase*]] [*Signal_shape*] + [**DISTOF1** *F1mag* [*F1phase*]] [**DISTOF2** *F2mag* [*F2phase*]]	SPICE3

Examples	
VSC 2 6 **AC** 1 mV 45.0	
VNP 4 7 **DC** 3V **AC** 1mV 90.0	
VPULSE 3 0 **PULSE** (0 5.2V 5ns 1ns 1ns 30ns 50ns)	
V6 2 0 **DC** 1V **AC** 1mV **SIN**(0 0.001 1MEG)	PSPICE
V2 5 0 **AC** 10mV **DISTOF1 DISTOF2** 0.001	SPICE3
V3 6 9 **AC** 0.1 45 **SFFM** (0 1 250kHz 0.5 3kHz)	SPICE3

Value is the dc and transient analysis value of the voltage source. If the source value is time-invariant (e.g., a power supply), then the value may optionally be preceded by the letters **DC**.

Mag is the ac magnitude and *Phase* is the ac phase. The source is set to this value in the ac analysis. If *Mag* is omitted following the keyword **AC**, a value of unity is assumed. If *Phase* is omitted, a value of zero is assumed. If parameters other than source values are omitted or set to zero, the default values shown will be assumed. If a source is assigned a time-dependent value, the time-zero value will be used for dc analysis.

The keyword STIMULUS is used in newer versions of PSPICE to call up custom signal shapes created with the stimulus editor. By specifying *Signal_shape*, a time-dependent waveform for transient analysis can be assigned. If a source is assigned a time-dependent value, the time-zero value is used for dc analysis. There are five independent source functions: pulse, exponential, sinusoidal, piecewise linear, and single-frequency FM. These five signal shapes are described in more detail in what follows.

1. Pulse waveforms

Form **PULSE** (V1 V2 TD TR TF PW PER)

Examples: VIN 3 0 **PULSE**(0 5V 10us 2us 2us 50us 100us)
VIN 5 0 **PULSE**(-5V 5V 10us 2us 2us 50us)

Parameters	*Meaning*	*Default*	*Units*
V1	Initial value	-	V
V2	Pulsed value	-	V
TD	Delay time	0.0	sec
TR	Rise time	*Tstep*	sec
TF	Fall time	*Tstep*	sec
PW	Pulse width	*Tstop*	sec
PER	Period	*Tstop*	sec

Parameters *Tstep* and *Tstop* are specified in **.TRAN** statement.



2. Sine waves

Form **SIN**(VO VA FREQ TD DF PHASE)

Examples: V3 3 0 **SIN**(1V 2V 10MEG 1ns 1MEG)
V6 6 2 **SIN**(0 10mV 100kHz)

parameters	*meaning*	*default*	*units*	*Restrictions*
VO	Offset		V	
VA	Amplitude		V	
FREQ	Frequency	1/*Tstop*	Hz	
TD	Delay	0.0	s	
DF	Damping factor	0.0	1/s	
PHASE	Phase	0.0	degree	PSPICE



$$\mathbf{V_S} = \mathbf{VA}\ \sin\left(2\pi\,\frac{\mathbf{PHASE}}{360}\right) \qquad \textbf{(V-1)}$$

The shape of the waveform is described by the following equations:

For time < **TD:**

$$\mathbf{VO} + \mathbf{VA}\sin\left[2\pi\left(\frac{\mathbf{PHASE}}{360}\right)\right] \qquad \textbf{(V-2)}$$

For time > **TD:**

$$\mathbf{VO} + \mathbf{VA}\exp\left[-\left(time-\mathbf{TD}\right)\mathbf{THETA}\right]\sin\left[2\pi\left(\mathbf{FREQ}\left(time-\mathbf{TD}\right)+\frac{\mathbf{PHASE}}{360}\right)\right] \qquad \textbf{(V-3)}$$

3. Exponential waveforms

Format **EXP**(V1 V2 TD1 TAU1 TD2 TAU2)

Examples: V3 5 0 **EXP**(-5 -3 5ns 20ns 30ns 30ns)
V6 3 2 **EXP**(0 5 5ns 20ns)

Parameters	*Meaning*	*Default*	*Units*
V1	Initial value		V
V2	Pulsed value		V
TD1	Rise delay time	0.0	s
TAU1	Rise time constant	*Tstep*	s
TD2	Fall delay time	TD1+*Tstep*	s
TAU2	Fall time constant	*Tstep*	s



The shape of the waveform is described by the following equations:

For *time* < **TD1**

$$v(time) = 0 \quad \textbf{(V-4)}$$

For **TD1** < *time* < **TD2**

$$v(time) = \mathbf{V1} + (\mathbf{V2} - \mathbf{V1})\left[1 - \exp\left(-\frac{time - \mathbf{TD1}}{\mathbf{TAU1}}\right)\right] \quad \textbf{(V-5)}$$

for *time* > **TD2**

$$v(time) = \mathbf{V1} + (\mathbf{V2} - \mathbf{V1})\left[1 - \exp\left(-\frac{time - \mathbf{TD1}}{\mathbf{TAU1}}\right)\right] + (\mathbf{V1} - \mathbf{V2})\left[1 - \exp\left(-\frac{time - \mathbf{TD2}}{\mathbf{TAU2}}\right)\right] \textbf{(V-6)}$$

4. Piecewise linear waveforms

Format: **PWL**(T1 V1 [Tn Vn] …)

Examples: V5 3 7 **PWL**(0 -7V 5us -7V 6us 5V 20us 5V 21us -7V 30us -8V)
V3 4 2 **PWL**(5ms 5V 20ms 5V 30ms 0V)

Parameters	*Meaning*	*Default*	*Units*
Tn	Time at corner	-	s
Vn	Voltage at corner	-	V

Each pair of values (**Tn**, **Vn**) specifies the value of the source **Vn** ([in V) at *time*=**Tn**. The value of the source at intermediate values of time is determined by using linear interpolation of the input values.



5. FM waveforms

Form **SFFM**(VO VA FC MDI FS)

Examples: **V**7 7 0 **SFFM**(0 1mV 20kHz 5 1kHz)
V1 12 0 **SFFM**(0 10V 300kHz 0.5 10kHz)

parameters	*meaning*	*default*	*units*
VO	Offset	-	V
VA	Amplitude	-	V
FC	Carrier frequency	*1/Tstop*	Hz
MDI	Modulation index	0	
FS	Signal frequency	*1/Tstop*	Hz



The SFFM (single-frequency frequency-modulated) waveform is described by the following equation:

$$v(time) = \mathbf{VO} + \mathbf{VA}\ \sin\left[2\pi\, \mathbf{FC}\, time + \mathbf{MDI}\ \sin\left(2\pi\, \mathbf{FS}\, time\right)\right] \qquad \textbf{(V-7)}$$

W - Current-Controlled Switch



General forms	*Restrictions*
W*name Pnode Nnode Vname model*	PSPICE
W*name Pnode Nnode Vname model* [**ON**] [**OFF**]	SPICE3

Examples	
W3 4 7 **V**IN WMOD	
WON 4 7 **V**ON WRELAY	
w1 4 0 **v**clock Switch	
W2 5 2 **V**R SM1 **ON**	SPICE3
wreset 7 2 **V**5 Lossysw **OFF**	SPICE3

The name of a current-controlled switch begins with the letter **W**. *Pnode* and *Nnode* represent the connections to the switch terminals. The model name, *Model*, is mandatory, while the initial conditions are optional. The controlling current is the current through the specified voltage source, defined the same way as in current-controlled dependent voltage (letter **F**) and current (letter **H**) sources. In SPICE3 the optional parameter **ON** or **OFF** specifies the switch state for the dc operating point.

The switch model allows an almost ideal switch to be described in SPICE. The switch is not quite ideal, in that the resistance cannot change from 0 to infinity, but must always have a finite and nonzero positive value. By proper selection of the on and off resistances, they can be effectively zero and infinity in comparison to other circuit elements.

The switch can have a hysteresis described by the **IH** parameter. For example, the current-controlled switch will be in the on state, with a resistance **RON**, at **IT**+**IH**. The switch will be in the off state, with a resistance **ROFF**, at **IT**-**IH**.

1. CSW model

.MODEL *Model* **CSW** [*list_of_parameters*] SPICE3

2. Model parameters

Name	*Parameter*	*Default*	*Units*	*Restrictions*
IT	Threshold current	0.0	A	SPICE3
IH	Hysteresis current	0.0	A	SPICE3
RON	On resistance	1.0	Ω	
ROFF	Off resistance	1/**GMIN**	Ω	SPICE3

Examples:

.MODEL SMOD **CSW** **IT**=3V **IH**=1V
.MODEL SMOD **CSW** **RON**=50 **ROFF**=10MEG **IT**=3mA

.MODEL *Model* **ISWITCH** [*list_of_parameters*] PSPICE

Model Parameters

Name	*Parameter*	*Default*	*Units*	*Restrictions*
ION	Threshold current for on state	10^{-3}	A	PSPICE
IOFF	Threshold current for off state	0.0	A	PSPICE
RON	On resistance	1.0	Ω	
ROFF	Off resistance	10^{6}	Ω	PSPICE

The use of an ideal element that is highly nonlinear, such as a switch, can cause large discontinuities to occur in the circuit node voltages. The rapid voltage change associated with a switch changing state can cause numerical round off or tolerance problems leading to erroneous results or time step difficulties.

You can improve the situation by taking the following steps: Set the switch impedances only high and low enough to be negligible with respect to other elements in the circuit. Using switch impedances that are close to "ideal" under all circumstances will aggravate the discontinuity problem. When modeling real devices such as MOSFETS, the on resistance should be adjusted to a realistic level depending on the size of the device being modeled.

If a wide range of **ON** to **OFF** resistance must be used (**ROFF/RON** $>10^{+12}$), then the tolerance on errors allowed during the transient analysis should be decreased by specifying the **.OPTIONS TRTOL** parameter to be less than the default value of 7.0. When switches are placed around capacitors, the **.OPTIONS CHGTOL** parameters should also be reduced. Suggested values for these two options are 1.0 and 1E-16, respectively. These changes inform SPICE to be more careful near the switch points so that no errors are made due to the rapid change in the circuit response.

X - Subcircuit Calls



General forms	*Restrictions*
X*name n1* [*n2 n3* **...**] *Subname*	
Examples	
X1 2 4 17 3 1 MULTI	
X7 3 2 6 7 OPAMP	
X2 3 2 6 7 1 NAND	

Subcircuits are specified in SPICE by using pseudo-elements beginning with the letter **X**, followed by the circuit nodes *n1* [*n2 n3* **...**] to be used in expanding the subcircuit and the name *Subname* of the subcircuit. The subcircuit is defined by the **.SUBCKT** statement. The number of nodes in the subcircuit call (**X** statement) must be the same as in the subcircuit declaration (**.SUBCKT**).

Z - MESFET

SPICE3 only



General forms	*Restrictions*
Z*name Dnode Gnode Snode Model* [*Rarea*] [**OFF**] [**IC**=*Vds, Vgs*]	SPICE3
Examples	
Z2 5 3 0 ZPMOD **OFF**	SPICE3
Z7 6 3 2 ZNMO **IC**= 5.0, 1.0	SPICE3

A MESFET is described by a statement starting with name of the MESFET device **Z***name*. This name must start with the letter **Z**. Node numbers *Dnode*, *Gnode*, and *Snode* for drain, gate, and source follow the name. Next, the model name *Model* is listed. Model parameters are specified in the **.MODEL** statement. The keywords **NMF** and **PMW** are used there for *n*-channel and *p*-channel respectively. *Rarea* is the relative area factor. If *Rarea* is not specified, 1 is assumed. An optional parameter **IC**=*Vd* is used together with a **UIC** in transient analysis. The keyword **OFF** indicates an optional starting condition for dc analysis.

1. MESFET models

.MODEL *Model_name* **NMF** [*Model parameters*]
.MODEL *Model_name* **PMF** [*Model parameters*]

2. Model parameters

Name	*Parameter*	*Units*	*Default*	*Typical*	*Rarea*
VTO	Pinch-off voltage	V	-2.0	-2.0	
BETA	Transconductance parameter	A/V	1.0e-4	1.0e-3	*
B	Doping tail extending parameter	1/V	0.3	0.3	*
ALPHA	Saturation voltage parameter	1/V	2	2	*
LAMBDA	Channel-length modulation parameter	1/V	0	1.0e-4	
RD	Drain ohmic resistance	Ω	0	100	*
RS	Source ohmic resistance	Ω	0	100	*
CGS	Zero-bias *G-S* junction capacitance	F	0	5 pF	*
CGD	Zero-bias *G-D* junction capacitance	F	0	5 pF	*
PB	Gate junction potential	V	1	0.6	
KF	Flicker noise coefficient	-	0	-	
AF	Flicker noise exponent	-	1	-	
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5	-	

Asterisks in the last column indicates that this parameter in all equations is multiplied by *Rarea* parameter specified in the **Z** device line.